



Teoría

ABAP Cloud Developer Extensibility – Acciones

SAP S/4HANA Cloud – Modelo de extensibilidad Clean Core





Contenido

1. ABAP Cloud Developer Extensibility – Acciones	3
1.1. Requerimiento – Extensión RAP Nueva Acción	3
1.2. Behavior Definition Root – Creación Extensión	7
1.3. Behavior Definition Root – Extensión Nueva Acción	12
1.4. Extensión CDS Interfaz – Elementos Requeridos	14
1.5. Acción – Implementación Behavior Pool	19
1.6. Behavior Definition Projection – Habilitar Extensión	23
1.7. Metadata Extensions – Exponer Acción en UI	25
1.8. Campo Solo Lectura	28



1. ABAP Cloud Developer Extensibility – Acciones

1.1. Requerimiento – Extensión RAP Nueva Acción


En esta lección se aborda la extensión de una aplicación estándar mediante la adición de un nuevo botón con una nueva acción y funcionalidad. El objetivo es ampliar las capacidades estándar de la aplicación para satisfacer un requerimiento empresarial específico.

Al seleccionar cualquier registro existente, las acciones o botones se habilitan. Se debe agregar visualmente el botón y realizar todas las extensiones necesarias de los objetos para disponer de la funcionalidad requerida.

Identificación de Peticiones:

En la pestaña Network en las herramientas de desarrollador. Las cuales se pueden habilitar en el navegador al estar en la aplicación requerida a modificar de la siguiente manera:

- Al presionar la tecla **F12**.
- Al presionar el conjunto de teclas **Ctrl + Shift + I**.
- Al seguir la siguiente ruta: Presionar el botón al lado derecho del ícono de la cuenta de Google si se encuentra en el navegador de Google Chrome. Luego seleccionar “More Tools” y por último “Developer Tools”.



Al presionar el botón “Go” en la aplicación se mostrarán las peticiones de tipo batch.

Identificación del Servicio y Service Binding:

Se identifican las peticiones tipo batch a un servicio OData. Los detalles de estas peticiones se encuentran en la sección headers. El sección “Request URL” tenemos información relevante como:

Name	Headers	Payload	Preview	Response	Initiator	Timing	Cookies
\$batch?sap-client=100	<input checked="" type="checkbox"/> General <input type="checkbox"/> Headers						
Request URL:	https://23bf02f0-9d4d-4fe7-ae57-cab54ffafb1e.abap-web.us.10.hana.ondemand.com/sap/opu/odata4/cdm/ui_agency_o4/srv/dmo/ui_agency/0001/\$batch?sap-client=100						
Request Method:	POST						
Status Code:	200 OK						
Remote Address:	52.23.1.21:443						
Referrer Policy:	strict-origin-when-cross-origin						

- **URL Relativa:** Identificar la URL relativa que apunta al servicio invocado.



General

Request URL:	https://23bf02f0-9d4d-4fe7-ae57-cab54ffaf1e.abap-web.us10.hana.ondemand.com/sap/opu/odata4/dmo/ui_agency_o4/srv/dmo/ui_agency/0001/\$batch?sap-client=100
Request Method:	POST
Status Code:	200 OK
Remote Address:	52.23.1.211:443
Referrer Policy:	strict-origin-when-cross-origin

Response Headers

Raw

- Servicio Odata:** El nombre del servicio Odata a la que se le hizo la llamada.

General


Request URL:	https://23bf02f0-9d4d-4fe7-ae57-cab54ffaf1e.abap-web.us10.hana.ondemand.com/sap/opu/odata4/dmo/ui_agency_o4/srv/dmo/ui_agency/0001/\$batch?sap-client=100
Request Method:	POST
Status Code:	200 OK
Remote Address:	52.23.1.211:443
Referrer Policy:	strict-origin-when-cross-origin

- Service Binding:** El nombre del service binding.

General

Request URL:	https://23bf02f0-9d4d-4fe7-ae57-cab54ffaf1e.abap-web.us10.hana.ondemand.com/sap/opu/odata4/dmo/ui_agency_o4/srv/dmo/ui_agency/0001/\$batch?sap-client=100
Request Method:	POST
Status Code:	200 OK
Remote Address:	52.23.1.211:443
Referrer Policy:	strict-origin-when-cross-origin

Con el nombre del service binding se puede acceder a este recurso del backend a través de Eclipse. Utilizando el ABAP Development Object, en donde se recomienda coloca el nombre del servicio entre “*” de tal manera que permita buscar en un rango más amplio el archivo.




Identificación del paquete de desarrollo de la aplicación:



Para identificar el paquete de desarrollo, se puede navegar desde el Service Binding hasta el Service Definition. Una vez allí, posicionar el cursor en cualquier línea del código y hacer clic derecho para abrir el menú de opciones.

```
@EndUserText.label: 'Agency'
@AbapCatalog.extensibility.extensible: true
@ObjectModel.leadingEntity.name: '/DMO/C_AgencyTP'
define service /DMO/UI_AGENCY
{
  expose /DMO/C_AgencyTP as Agency;
}
```


Al seleccionar la opción “Change Package Assignment...”.



Se puede identificar el paquete actual del proyecto de la aplicación.




Para posteriormente añadir el paquete en favoritos.





Donde se puede analizar todos los objetos del proyecto para añadir la extensión de la funcionalidad.



1.2. Behavior Definition Root – Creación Extensión

Para la creación de una nueva acción o botón en una aplicación estándar de tipo RAP, es necesario crear una extensión de comportamiento basado en el Behavior Definition de la entidad raíz permitiendo agregar nuevas funcionalidades a una aplicación estándar.

Identificación del Desarrollo y Artefactos a Extender:

Se debe navegar desde el Service Binding, pasando por el Service Definition hasta la entidad de consumo o de proyección y luego a la entidad de tipo raíz. A través de esto, se identifica la definición del comportamiento correspondiente.

The screenshot shows the SAP IDE Workbench with the following details:


- Project Explorer:** Shows the project structure with various packages and components.
- Code Editor:** Displays the following ABAP code, which defines a root view entity for the /DMO/R_AGENCYTP entity.

```

1 *@AccessControl.authorizationCheck: #NOT_REQUIRED
2 @EndUserText.label: 'Agency'
3 @Search.searchable: true
4
5 @AbapCatalog.extensibility: {
6     extensible: true,
7     elementSuffix: 'ZAG',
8     allowNewDatasources: false,
9     dataSources: ['_Extension'],
10    quota: {
11        maximumFields: 500,
12        maximumBytes: 50000
13    },
14    allowNewCompositions: true
15 }
16
17 define root view entity /DMO/R_AGENCYTP
18 as select from /DMO/I_Agency as Agency
19 association [0..1] to /I_Country as _Country on $projection.CountryCode = _Country.Cou
20 association [1] to /DMO/E_Agency as _Extension on $projection.AgencyID = _Extension.Agency

```

El cual posee el mismo nombre que la entidad raíz.



Las acciones se definen en el Behavior Definition de la entidad raíz de la aplicación. Estas acciones son las funcionalidades básicas y esenciales que la aplicación puede ejecutar. Dicha entidad debe poseer características de extensión implementadas de la siguiente manera:

```
managed implementation in class /dmo/bp_r_agencytp unique;
strict ( 2 );
with draft;
extensible
{
    with determinations on modify;
    with determinations on save;
    with validations on save;
    with additional save;
}

•define behavior for /DMO/R_AgencyTP alias /DMO/Agency
persistent table /dmo/agency
draft table /dmo/agency_d query /DMO/R_AgencyDraft
lock master
total etag LastChangedAt
authorization master ( global )
etag master LocalLastChangedAt
late numbering
extensible
```

Además es necesario que la aplicación tenga implementada un Behavior Definition de una entidad de interfaz. Esta entidad permite que los desarrolladores amplíen y mejoren los objetos de negocio sin alterar la lógica subyacente. Basada en la entidad raíz (Root Entity), la entidad de interfaz integra anotaciones y proyecciones necesarias para gestionar datos de manera eficiente y coherente. La cual actúa como una puerta que permite implementar nuevas funcionalidades en la aplicación. Si la interfaz posee características de extensión, se puede utilizar dicha interfaz como "puerta" para introducir nuevas acciones en el nodo principal de la aplicación.

```
interface;
use draft;
extensible;

define behavior for /DMO/I_AgencyTP alias /DMO/Agency
{
    use create;
    use update;
    use delete;

    use action Resume;
    use action Edit;
    use action Activate;
    use action Discard;
    use action Prepare;
}
```



Los elementos necesarios para las extensiones deben estar presentes de igual forma en el behavior definition de consumo. Esto garantiza que la aplicación pueda manejar las nuevas funcionalidades que queremos agregar.

```

projection;
strict(2);
use draft;
extensible;

define behavior for /DMO/C_AgencyTP alias /DMO/Agency
extensible
{
    use create;
    use update;
    use delete;

    use action Resume;
    use action Edit;
    use action Activate;
    use action Discard;
    use action Prepare;
}

```

En una configuración habitual, los elementos necesarios para las extensiones están presentes tanto en el nodo raíz (Root) como en la parte de consumo de datos. Esto garantiza que la aplicación pueda manejar las nuevas funcionalidades que requieran agregar.


Proceso de Creación de un Behavior Extension:

Para crear una extensión de comportamiento, es necesario localizar la definición de comportamiento (Behavior Definition) de la entidad raíz dentro del paquete de la aplicación. Luego, haz clic derecho en el nombre del objeto para desplegar el menú de opciones y seleccionar "New Behavior Extension".







En la siguiente ventana es fundamental colocar el paquete Z donde se estarán realizando los desarrollos de las extensiones. También se puede validar que el nombre del Behavior Definition de la entidad raíz se encuentra ya asignado.




Luego es necesario especificar el Behavior Definition de la entidad interfaz. Utilizando el botón “Browse” en el campo BO: Interface.



Se puede seleccionar rápidamente el Behavior Definition de la entidad interfaz más rápidamente.



Para finalizar con el proceso presionando el botón “Next” para agregar los cambios realizados en el paquete en una orden de transporte.



El Behavior extensión se mostrará de la siguiente manera:

```
extension using interface /dmo/i_agencytp
implementation in class zbp_r_agencytp unique;
```

1.3. Behavior Definition Root – Extensión Nueva Acción

La extensión de una aplicación estándar de tipo RAP mediante la creación de una nueva acción o botón en la interfaz de usuario implica la definición y personalización del comportamiento de la aplicación. Este proceso incluye la identificación de los artefactos necesarios dentro del paquete de desarrollo, así como la definición de nuevas acciones utilizando nodos y alias existentes. Al utilizar extensiones en la definición de comportamiento (Behavior Definition), se pueden agregar acciones adicionales que modifican o amplían las funcionalidades estándar de la aplicación. La interfaz y las extensiones de metadatos



(metadata extensions) juegan un papel crucial en habilitar y visualizar estas nuevas funcionalidades, asegurando que las acciones añadidas se integren de manera coherente en la aplicación estándar.

Continuando con el proceso, después de crear la extensión de comportamiento (Behavior Extension), se puede verificar que la relación está disponible dentro de la definición de comportamiento (Behavior Definition) de la entidad raíz. Para hacerlo, selecciona el símbolo ubicado en la línea donde se encuentra la sentencia **define behavior for**. al desplegar el menú de relaciones.

```

1 managed implementation in class /dmo/bp_r_agencytp unique;
2 strict ( 2 );
3 with draft;
4 extensible
5*{
6   with determinations on modify;
7   with determinations on save;
8   with validations on save;
9   with additional save;
10 }
11
12•define behavior for /DMO/R_AgencyTP alias /DMO/Agency
13 persistent table /dmo/agency
14 draft table /dmo/agency_d query /DMO/R_AgencyDraft
15 lock master
16 total etag LastChangedAt
17 authorization master ( global )
18 etag master LocalLastChangedAt
19 late numbering
20 extensible

```

Luego es necesario seleccionar ubicado en la línea donde se encuentra la sentencia **define root view entity**, para desplegar el menú de relaciones de la entidad de consumo.

```

1 managed implementation in class /dmo/bp_r_agencytp unique;
2 strict ( 2 );
3 with draft;
4 extensible
5*{
6   with determinations on modify;
7   with determinations on save;
8   with validations on save;
9   with additional save;
10 }
11
12•define root view entity /DMO/Agency
13
14 Extended with
15
16
17

```

En donde es importante saber, que se debe utilizar el alias del nodo que se requiere extender dentro de la definición del comportamiento en de



la entidad raíz. Y se procede a la definición de la nueva acción en la extensión de comportamiento.

```
managed implementation in class /dmo/bp_r_agencytp unique;
strict ( 2 );
with draft;
extensible
{
    with determinations on modify;
    with determinations on save;
    with validations on save;
    with additional save;
}

•define behavior for /DMO/R_AgencyTP alias [/DMO/Agency]
persistent table /dmo/agency
draft table /dmo/agency_d query /DMO/R_AgencyDraft
lock master
total etag LastChangedAt
authorization master ( global )
etag master LocalLastChangedAt
late numbering
extensible
```

Recordando que se utilizará el alias del nodo y no el nombre de la fuentes de datos a pesar de que en este caso sean iguales.

```
extension using interface /dmo/i_agencytp
implementation in class zbp_r_agencytp unique;

extend behavior for [/DMO/Agency]

{
```

Luego se procede a definir la acción para luego seleccionar una autorización para la acción. Las posibles autorizaciones EML que se establecen dentro del Behavior Definition se explican a profundidad en las documentaciones de RAP, disponibles en su curso o máster correspondiente en la plataforma de Logali.

```
extension using interface /dmo/i_agencytp
implementation in class zbp_r_agencytp unique;

extend behavior for /DMO/Agency

•{
    action (au)
}
```



En este caso no se establece ninguna autorización con la sentencia **authorization : none**. Para luego definir el nombre de la acción, donde es obligatorio colocar el prefijo “zz” al nombre debido a que la definición es un elemento personalizado. Seguido de la sentencia **result [1] \$self;** para finalizar con la instrucción.

```
extension using interface /dmo/i_agencytp
implementation in class zbp_r_agencytp unique;

extend behavior for /DMO/Agency

{
    action (authorization : none) zzAssignType result [1] $self;
}
```

1.4. Extensión CDS Interfaz – Elementos Requeridos

Este proceso se centra en la extensión de una aplicación estándar de tipo RAP mediante la creación de una nueva acción o botón en la interfaz de usuario. La extensión implica la definición de lógica y la implementación de funcionalidades adicionales que se ejecutarán cuando el usuario interactúe con la nueva acción.

Dado que la entidad de tipo interfaz permite implementar nuevas funcionalidades sin alterar la lógica subyacente. Y como dicha entidad selecciona los datos de la entidad raíz que ya tiene el campo personalizado agregado, se debe proyectar dicho campo en la entidad de interfaz a través de una extensión. Esto permitirá utilizarlo en la lógica del código fuente en el behavior pool, explicación que se realizará más adelante en el desarrollo del documento, dando vida a este método.

```
define root view entity /DMO/I_AgencyTP
    provider contract transactional_interface
    as projection on /DMO/R_AgencyTP as Agency
{
    @ObjectModel.text.element: [ 'Name' ]
    key AgencyID,
    *
    @Search.defaultSearchElement: true
    @Search.fuzzinessThreshold: 0.8
    @Semantics.text: true
    @Semantics.organization.name: true
    Name,
```

Esto se puede validar navegando desde la entidad interfaz a la entidad raíz.



```
•@AccessControl.authorizationCheck: #NOT_REQUIRED
@EndUserText.label: 'Agency'
@Search.searchable: true

@AbapCatalog.extensibility: {
  extensible: true,
  elementSuffix: 'ZAG',
  allowNewDatasources: false,
  dataSources: ['_Extension'],
  quota: {
    maximumFields: 500,
    maximumBytes: 50000
  },
  allowNewCompositions: true
}

define root view entity /DMO/R_AgencyTP
  as select from /DMO/I_Agency as Agency
  association [0..1] to /DMO/I_Country as _Country on $projection.CountryCode = _CountryCode
  association [1] to /DMO/E_Agency as _Extension on $projection.AgencyID = _ExtensionID
```

Y ejecutándola al presionar la tecla **F8** para ver su contenido.

Dado a que no es posible utilizar la lectura del nodo raíz directamente en la interfaz de usuario, se debe recurrir a la entidad interfaz habilitada en la definición de comportamiento asociada.

```
interface;
use draft;
extensible;

define behavior for /DMO/I AgencyTP alias /DMO/Agency
{
    use create;
    use update;
    use delete;

    use action Resume;
    use action Edit;
    use action Activate;
    use action Discard;
    use action Prepare;
}
```

Donde al navegar a dicha entidad de interfaz y pulsar F8, se puede comprobar que el campo no está disponible.



```

@Metadata.ignorePropagatedAnnotations: true
@EndUserText.label: 'Agency'
@Search.searchable: true
@AccessControl.authorizationCheck: #NOT_REQUIRED

@AbapCatalog.extensibility: {
    extensible: true,
    dataSources: [ 'AGENCY' ],
    elementSuffix: 'ZAG',
    quota: {
        maximumFields: 500,
        maximumBytes: 50000
    }, allowNewCompositions: true
}


define root view entity /DMO/I_AgencyTP
    provider contract transactional_interface
    as projection on /DMO/R_AgencyTP as Agency

```


Ya que por el momento no ha sido agregado.

Data Preview											
Agency	ID	Name	PostalCode	City	Country	PhoneName	Database	Variable	Attachment	MinExt	File
Your Choice	Galleria No.	94545	Nordstrasse 10	DE		+49 1516-5545-0	info@yourchoice.de	http://www.yourchoice.de		0.000000	
Bella Italia	Via Mancio	01123	224 Balmi	IT		+39 010-467123	info@bella-italia.it.itv	http://www.bella-italia.it		0.000000	
Hot Socks		80539	Sydney	AU		+61 2 2004 5000	info@hot-socks.com.au	http://www.hot-socks.com.au		0.000000	
Al's Base	123 Main St.	100028	Santa Fe	US		+1 505-555-1234	info@alsbase.com	http://www.alsbase.com		0.000000	
Horizon Support	42100		Neumarkt	AT		+43 7361 8963	info@horizon-support.at	http://www.horizon-support.at		0.000000	
Travel Inc.	Altman S.	10557	Berlin	DE		+49 30-622060	info@travel-inc.de	http://www.travel-inc.de		0.000000	
Progeny	123 Main St.	100028	Stockholm	SE		+46 8 555-2345	info@progeny.se	http://www.progeny.se		0.000000	
Blue Planet	74 Court	811707N	London	GB		+44 131 234-2344	info@blueplanet.com	http://www.blueplanet.com		0.000000	
Pizzaplex	Gold Park 4	80008	Zuerich	CH		+41 1 345-5531	info@pizzaplex.ch	http://www.pizzaplex.ch		0.000000	
Kangaroo	Lancaster	20003	London	GB		+44 171-2076768	info@kangaroo.sap	http://www.kangaroo.sap		0.000000	
Horizon Support	42100		London	GB		+44 171-2076769	info@horizon-support.sap	http://www.horizon-support.sap		0.000000	
All's Base	45 Mac A.	191113	Boston	US		+1 508-652-5200	info@alls-base.com	http://www.alls-base.com		0.000000	
Super Agent	50 Crane	912184G	Glasgow	GB		+44 141 715643	info@superagent.sap	http://www.superagent.sap		0.000000	
Wang Cho	Gagarine	190001	Moscow	RU		+7 951-271201	info@wang-cho.sap	http://www.wang-cho.sap		0.000000	
Answer It	Adler Pl.	40023	Hannover	DE		+49 511-857500-0	info@answer-it.sap	http://www.answer-it.sap		0.000000	


Por lo que se necesita crear una vista de extensión para proyectar el campo. Como se hizo anteriormente con la entidad raíz.



Se crea el nuevo Data Definition colocando el paquete Z donde se estarán agregando las ampliaciones al programa estándar por medio de extensiones, un nombre para el objeto donde por lo general se utiliza el sufijo “zz”. No obstante, en este apartado se debe emplear la nomenclatura específica utilizada por el cliente en su organización. Además, se debe colocar el nombre de la entidad raíz en el campo “Referenced Object”.



Recordando seleccionar la plantilla **extendViewEntity** que se encuentra en la carpeta **View Extend (creation)**.



Donde para proyectar el campo desde la entidad raíz a la entidad interfaz, es necesario utilizar el nombre del alias de la entidad raíz.



```

    }, allowNewCompositions: true
}

define root view entity /DMO/I_AgencyTP
provider contract transactional_interface
as projection on /DMO/R_AgencyTP as Agency
{
    @ObjectModel.text.element: ['Name']
key AgencyID,
    @Search.defaultSearchElement: true
    @Search.fuzzinessThreshold: 0.8
    @Semantics.text: true
    @Semantics.organization.name: true
    Name,
    @Semantics.address.street: true
    Street,
    @Semantics.address.zipCode: true
    PostalCode,

```

Y especificar el campo proyectado.

```

extend view entity /DMO/I_AgencyTP with {
    Agency.zzagtypezag
}

```

1.5. Acción – Implementación Behavior Pool

Las funcionalidades adicionales que se ejecutarán en la interfaz de usuario cuando el usuario interactúe con un nuevo botón. Conllevan en su definición la implementación de la lógica que se ejecutará en la nueva acción a través de una clase de tipo Behavior Pool, que extiende la definición de comportamiento raíz de una aplicación estándar de tipo RAP.

La lógica de las acciones de una aplicación RAP se desarrollan en la clase Behavior Pool implementada en el Behavior Definition de la entidad raíz. Pero debido a que la lógica de la nueva acción personalizada no se puede agregar directamente en la clase, se hace uso de la extensión de comportamiento.



```

managed implementation in class /dmo/bp_r_agencytp unique;
strict ( 2 );
with draft;
extensible
{
    with determinations on modify;
    with determinations on save;
    with validations on save;
    with additional save;
}

define behavior for /DMO/R_AgencyTP alias /DMO/Agency
persistent table /dmo/agency
draft table /dmo/agency_d query /DMO/R_AgencyDraft
lock master
total etag LastChangedAt
authorization master ( global )
etag master LocalLastChangedAt
late numbering
extensible

```

Creación de la Clase Behavior Pool de extensión :

En la carpeta **Core Data Services** del paquete de desarrollo "Z" donde se han estado agregando las ampliaciones al programa estándar a través de extensiones. En esta sección, se selecciona la extensión de comportamiento correspondiente.



Luego se selecciona el nombre propuesto para la nueva clase Behavior Pool, que se encuentra al finalizar la sentencia **implementation in class**.

```

extension using interface /dmo/i agencytp
implementation in class zbp_r_agencytp unique;

extend behavior for /DMO/Agency

{
    action (authorization : none) zzAssignType result [1] $self;
}

```

Se utiliza la opción de propuesta sobre el mensaje de advertencia para crear la clase. Al hacer doble clic en la opción “Create behavior implementation class”, se creará dicha clase finalizando el proceso al incorporar el artefacto en el paquete de desarrollo "Z" en una orden de transporte.



```

1 extension using interface /dmo/i_agencytyp
2 implementation in class zbp_r_agencytyp unique;
3
4 extend behavior for /DMO/
5
6•{
7   action (authorization
8 }

```

Create behavior implementation class zbp_r_agencytyp.
 Press 'Ctrl+Shift+1' to show in Quick Assist View

En la pestaña “Global Class” se hace referencia a la entidad raíz, verificando que la lógica añadida en esta clase de extensión “behavior pool” afectará dicha entidad.

```

1•class zbp_r_agencytyp definition
2   public
3   abstract
4   final
5   for behavior of /dmo/r_agencytyp.
6 endclass.
7
8•class zbp_r_agencytyp implementation.
9 endclass.

```

Global Class

Al posicionar el cursor encima del nombre de la acción se comprueba que la lógica de esta por medio de un método en dicha clase no ha sido implementada.

```

1 extension using interface /dmo/i_agencytyp
2 implementation in class zbp_r_agencytyp unique;
3
4 extend behavior for /DMO/Agency
5
6•{
7   action (authorization : none) zzAssignType result [1] $self;
8 }

```

Action /DMO/AGENCYTYP - ZZASSIGNTYPE is not implemented

Por lo que al hacer clic en el nombre de la acción y utilizar la ayuda del editor y seleccionar la opción “Add Method for action in new local...”, permitirá la implementación del método asociado a la acción.



```

1 extension using interface /dmo/i_agencytyp
2 implementation in class zbp_r_agencytyp unique;
3
4 extend behavior for /DMO/Agency
5
6{
7  action (authorization : none) zzAssignType result [1] $self;
8 }

```

Add method for action zzAssignType of entity /dmo/i_agencytyp in new local helper class

En la pestaña Local Types se puede comprobar que el método ha sido implementado.

```

1•CLASS lhc/_dmo/agency DEFINITION INHERITING FROM cl_abap_behavior_handler.
2
3  PRIVATE SECTION.
4
5    METHODS zzAssignType FOR MODIFY
6      IMPORTING keys FOR ACTION /DMO/Agency~zzAssignType RESULT result.
7
8  ENDCLASS.
9
10•CLASS lhc/_dmo/agency IMPLEMENTATION.
11
12  METHOD zzAssignType.
13  ENDMETHOD.
14
15 ENDCLASS.
16
17*** use this source file for the definition and implementation of
18 *** local helper classes, interface definitions and type
19 *** declarations
20

```

Y es dentro de ese método que se agregará la lógica asociada a la acción. Sin embargo en primera instancia no podemos utilizar la sentencia **in local mode**, porque realmente esta clase no se encuentra en el nodo de la entidad raíz.

```

method zzAssignType.

  read entities of /DMO/R_AgencyTP in local mode
    entity /DMO/Agency
    fields ( CountryCode zzagtypezag )
    with corresponding #( keys )
    result data(agencies).

endmethod.

```

Por lo tanto para eliminar la advertencia se debe eliminar la sentencia **in local mode** si se va a consultar directamente desde la entidad raíz.

Por otro lado, se puede utilizar la entidad de interfaz para realizar consultas, ya que la clase hace referencia al nodo correspondiente de dicha entidad. Esto permite el uso de la sentencia **in local mode**, mejorando significativamente el rendimiento de las consultas.



```
method zzAssignType.

    read entities of /DMO/I_AgencyTP in local mode
        entity /DMO/Agency
        fields ( CountryCode zzagtypezag )
        with corresponding #( keys )
        result data(agencies).

    loop at agencies assigning field-symbol(<agency>).

        case <agency>-CountryCode.
            when 'US'.
                <agency>-zzagtypezag = 'AM'.
            when others.
                <agency>-zzagtypezag = 'EU'.
        endcase.

    endloop.

endmethod.
```

Posteriormente agregar la lógica correspondiente que en este caso el objetivo será de que al seleccionar un registro permitirá cambiar el estado del campo Agency Type dependiendo del código del país.

```
method zzAssignType.

    read entities of /DMO/I_AgencyTP in local mode
        entity /DMO/Agency
        fields ( CountryCode zzagtypezag )
        with corresponding #( keys )
        result data(agencies).

    loop at agencies assigning field-symbol(<agency>).

        case <agency>-CountryCode.
            when 'US'.
                <agency>-zzagtypezag = 'AM'.
            when others.
                <agency>-zzagtypezag = 'EU'.
        endcase.

    endloop.

    modify entities of /DMO/I_AgencyTP in local mode
        entity /DMO/Agency
        update fields ( zzagtypezag )
        with corresponding #( agencies ).

endmethod.
```

Para luego utilizar el parámetro result para actualizar la interfaz de usuario.

```
modify entities of /DMO/I_AgencyTP in local mode
    entity /DMO/Agency
    update fields ( zzagtypezag )
    with corresponding #( agencies ).

    result = value #( for agency in agencies ( %tky = agency-%tky
        %param = agency ) ).

endmethod.
```



1.6. Behavior Definition Projection – Habilitar Extensión

El proceso para habilitar la acción en la interfaz de usuario comienza con la extensión de la definición de comportamiento de tipo projection. Esta extensión permite habilitar la nueva acción creada sobre la definición del comportamiento en la entidad de consumo.

La habilitación de las acciones se logra mediante la inclusión de la sentencia "**use action**" en la definición del comportamiento de la entidad de consumo o proyección. Esta acción puede validarse dentro de la propia definición del comportamiento. Además, se puede verificar que dicha definición permite realizar extensiones para habilitar nuevas acciones, mediante el uso de la instrucción extensible.

```
/DMO/R_AGENCYTP [TRU] Z-R_AGENCYTP [TRU] /DMO/C_AGENCYTP X
projection;
strict(2);
use draft;
extensible;

define behavior for /DMO/C_AgencyTP alias /DMO/Agency
extensible
{
    use create;
    use update;
    use delete;

    use action Resume;
    use action Edit;
    use action Activate;
    use action Discard;
    use action Prepare;
}
```


Creación de la Extensión de comportamiento para la entidad de consumo o proyección:

Para la creación de la extensión de comportamiento para la entidad de consumo se puede hacer clic derecho en el nombre de la definición en la carpeta "Core Data Services" del paquete de desarrollo Z donde se estarán realizando los desarrollos de las extensiones y seleccionar la opción "New Behavior Extension".





De manera que en la siguiente ventana se especifica el paquete Z, un nombre, descripción y por la acción realizada anteriormente ya se encuentra especificada en la definición de comportamiento de la entidad de consumo a la que hace referencia.



Generando el siguiente elemento de extensión.

```
extension for projection;
extend behavior for /DMO/Agency
{|
  }
```

La habilitación de la acción se logra mediante la agregación de la sentencia "use action" en la definición de la extensión. Se verifican y activan los cambios, asegurando que la implementación sea correcta y libre de advertencias. Es importante mencionar que el nombre de la acción se debe colocar manualmente debido a que el editor de texto no reconoce el nombre de la acción a extender para su habilitación. El nombre debe ser copiado desde la extensión de comportamiento basado en el Behavior Definition de la entidad raíz.

```
extension for projection;
extend behavior for /DMO/Agency
{
  use action zzAssignType;
}
```



1.7. Metadata Extensions – Exponer Acción en UI

Los Metadata Extensions son esenciales para personalizar y adaptar la interfaz de usuario a las necesidades específicas del requerimiento, mejorando así la funcionalidad y usabilidad de la aplicación. Permiten realizar cambios significativos sin comprometer la estabilidad del sistema, facilitando la actualización y mantenimiento de la UI. Por lo que se utilizará para habilitar y configurar un nuevo botón en la interfaz de una tabla, que permitirá modificar el tipo de agencia de acuerdo con la lógica implementada en el behavior pool.

Es por medio del Metadata extension extendida de la entidad de consumo creado con anterioridad que se puede implementar un nuevo botón en la interfaz de usuario y vincularlo a la acción que ejecuta la lógica agregada en la clase Behavior Pool, que extiende la definición de comportamiento de la entidad raíz.

```
@Metadata.layer: #CUSTOMER
annotate entity /DMO/C_AgencyTP with
{
    @UI: { lineItem: [ { position: 21, importance: #HIGH } ],
            fieldGroup: [ { position: 21, qualifier: 'General_FG' } ] }
}
zzagtypezag;
```

Vinculación del Botón en el Ítem de Línea:

Para agregar el botón en la interfaz, es necesario especificar por medio de la anotación **@UI.lineItem**, la posición utilizando la propiedad **position**. Generalmente, se establece en la posición 10 para que aparezca en primer lugar entre los botones del encabezado de la tabla en la interfaz de usuario. Además, se debe definir el tipo de acción que realizará el botón mediante la propiedad **type**, en este caso, **#FOR_ACTION**. Posteriormente, se vincula la propiedad **dataAction** con el nombre de la acción implementada, que en este ejemplo es



“zzAssignType”. Finalmente, se asigna una etiqueta al botón utilizando la propiedad **Label**, de modo que este sea el texto visible en el botón dentro de la interfaz de usuario.

```
@Metadata.layer: #CUSTOMER
annotate entity /DMO/C_AgencyTP with
{
    @UI: { lineItem: [ { position: 21, importance: #HIGH },
                      { position: 10,
                        type: #FOR_ACTION,
                        dataAction: 'zzAssignType',
                        label: 'Set Agency Type' } ],
           fieldGroup: [ { position: 21, qualifier: 'General_FG' } ] }
}
zzagtypezag;
```

Al ir a la aplicación en la interfaz de usuario y presionar el botón “Go” para mostrar los registros consultados de la capa de persistencia.

Se puede comprobar que el botón se encuentra habilitado en la interfaz de usuario pero deshabilitado.

Por otro lado al seleccionar un registro se habilitará y al presionar dicho botón ejecutará la lógica personalizada vinculada a la acción.

En este caso cambiar el campo “Agency Type”, dependiendo del código de país.

Sin embargo al seleccionar un registro de la tabla.



Y presionar el botón “Edit”.

Por el momento permitirá la modificación manualmente del campo personalizado agregado por medio de extensiones.


1.8. Campo Solo Lectura

La restricción de solo lectura (read-only) en campos específicos de la interfaz de usuario se utiliza para garantizar la integridad de los datos y limitar el acceso de edición a ciertos campos críticos. Esta funcionalidad se implementa mediante extensiones de metadata (metadata extensions), permitiendo que los campos no sean modificables por los usuarios, mientras se configuran acciones específicas a través de botones en la interfaz.

Para establecer el campo personalizado como no editable o campo de sólo lectura al momento de editar el registro.



Es necesario agregar el nuevo comportamiento del campo en la extensión de comportamiento de la entidad raíz. Recordando que para acceder a dicho elemento se debe acceder a la carpeta “Behavior Definitions” en el paquete Z donde se estarán realizando los desarrollos de las extensiones.



De manera que es necesario ampliar la definición desarrollada.

```

1 extension using interface /dmo/i_agencytp
2 implementation in class zbp_r_agencytp unique;
3
4 extend behavior for /DMO/Agency
5
6{
7   action ( authorization : none ) zzAssignType result [1] $self;
8 }
```

Agregando la instrucción **field**, con la propiedad **readonly**, seguido del nombre del campo que se establecerá como campo de solo lectura.

```

extension using interface /dmo/i_agencytp
implementation in class zbp_r_agencytp unique;

extend behavior for /DMO/Agency

{
  field ( readonly ) zzagtypezag;
  action ( authorization : none ) zzAssignType result [1] $self;
}
```

Además de que es posible colocar el botón asociado a la acción en el menú de edición. Por lo que para implementar dicha acción se debe agregar la anotación **@UI.identification**. Con las mismas propiedades establecidas en la anotación **@UI.lineItem**.



```
@Metadata.layer: #CUSTOMER
annotate entity /DMO/C_AgencyTP with
{
    @UI: { lineItem: [ { position: 21, importance: #HIGH },
                      { position: 10,
                        type: #FOR_ACTION,
                        dataAction: 'zzAssignType',
                        label: 'Set Agency Type' } ],
           fieldGroup: [ { position: 21, qualifier: 'General_FG' } ],
           identification: [ { position: 10,
                               type: #FOR_ACTION,
                               dataAction: 'zzAssignType',
                               label: 'Set Agency Type' } ] }
}
zzagtypezag;
```

Para comprobar los cambios realizados al seleccionar un registro en la tabla para su modificación al presionar el botón “Edit”.

The screenshot shows the SAP Fiori interface for editing an agency record. At the top, there's a header with the title 'modified agency name' and a 'Rating' section showing 4.7 out of 5 stars. Below the header is a toolbar with buttons for 'Edit', 'Delete', 'Set Agency Type', 'Create from Template', and a dropdown. The main area is divided into sections: 'General', 'Address', and 'Contact Data'. In the 'General' section, the 'Agency Name' field contains 'modified agency name'. The 'Address' section includes fields for 'Street' (Berliner Allee 11), 'City' (Duesseldorf), and 'Country/Region Key' (Germany (DE)). The 'Contact Data' section includes fields for 'Phone No.' (+49 2120 69555), 'Web Address' (<http://www.flyhigh.sap>), and 'E-Mail Address' (info@flyhigh.sap). The 'Agency Type' field is highlighted with a red box.

Se puede evidenciar que el campo no se pueda editar.

This screenshot shows the same SAP Fiori edit screen as the previous one, but with a different state. The 'Agency Type' field is now grayed out and cannot be edited, as indicated by the red box around it. The rest of the fields are in their original state.

Además de que en la visualización y edición del registro seleccionado aparecerá el botón con la acción personalizada.

This screenshot shows the SAP Fiori edit screen again. The 'Set Agency Type' button in the toolbar is highlighted with a red box. The rest of the screen, including the agency details and the 'Edit' button, appears normal.



Permitiendo de esta manera cambiar el estado del campo Agency Type dependiendo del código de país, desde el menú de edición.

The screenshot shows the 'modified agency name' screen in SAP S/4HANA Cloud. At the top, there is a navigation bar with icons for back, forward, search, and other functions. Below the navigation is a 'Rating' section with a logo, a star icon, and the text '4.7 out of 5'. The main content area is titled 'Agency' and contains three tabs: 'General', 'Address', and 'Contact Data'. The 'General' tab is selected. It includes fields for 'Agency Name' (set to 'modified agency name'), 'Slogan' (containing placeholder text), and 'Agency Type' (set to 'EU'). A red box highlights the 'Agency Type' field, and a mouse cursor is positioned over it. Other visible fields include 'Street' (Bertolt-Brecht-Allee 11), 'City' (Düsseldorf), 'Postal Code' (40990), 'Country/Region Key' (Germany (DE)), 'Phone No.' (+49 2122 60909), and 'Web Address' (http://www.flyhigh.sap). There are also 'Set Agency Type' and 'Create from Template' buttons at the top right of the form.